
Infrarood ontvanger nu nog 
eenvoudiger 
 
Peter van Overbeek 
 
PTC Print 
 
Scanned, ocr’ed and converted to PDF by HansO,  2001 
 
Nieuwe ontwikkelingen in de elektronica maken allerlei toepassingen keer op keer 
eenvoudiger en goedkoper. Zo bestaat er tegenwoordig een driepoot-IC, dat een infrarood 
detector samen met ingebouwde demodulator bevat: de SFH505A van Siemens. Het ding 
blijkt in de elektronicawinkel nog geen tientje te kosten. In PRINT 72 schreef Theo 
Maassen over een RC-5 infrarood ontvanger via UNIFACE, waarbij van dit IC gebruik 
wordt gemaakt. Het betreft een schakeling uit het blad Elektuur met een vijftal IC's, die 
op een UNIFACE EXPIO-print kan worden aangesloten. Maar het kan nog véél 
eenvoudiger. 
 
De SFH505A blijkt direct op de joystickingang van een MSX-computer aangesloten te 
kunnen worden. In PRINT 39 van juli 1990 stond ook al een schema en programma voor 
een infrarood ontvanger, aan te sluiten op de joystickingang van een MSX-computer. 
Daarin wordt gebruik gemaakt van een fotodiode BPW50 als infrarood detector en van 
het Philips-IC TDA3047 als demodulator. Daarbij horen nog twee filters, elk bestaande 
uit een spoeltje en een condensator. Deze moeten afgestemd worden op de 
modulatiefreguentie van ca 36 kHz, zoals voor infrarood afstandbedieningen gebruikelijk 
is. Het afregelen van deze filters is echter niet zo eenvoudig, zeker als u daarbij niet over 
een oscilloscoop kunt beschikken. Dit bleek dan ook bij een aantal nabouwers problemen 
op te leveren. Het nieuwe IC SFH505A vervangt zowel de fotodiode als de demodulator 
en de bijbehorende filters. Dit IC bevat digitale filters die niet meer afgeregeld behoeven 
te worden. Daarmee is het nabouwen van de infrarood ontvanger al een heel stuk 
eenvoudiger geworden. 
 

 
 
Fig. 1. Infrarood ontvanger met SFH 505A 
 



Nieuw schema 
 
Het oorspronkelijke schema uit PRINT 39 bevat, behalve de demodulator TDA3047, nog 
twee andere IC's: een 74LS04 en een NE556. Hiermee wordt het kloksignaal van de 
infrarood zender gereconstrueerd, als hulpmiddel bij het decoderen van de gebruikte RC-
5 code. Bovendien wordt bij ontvangst van infrarood signalen een startimpuls aan de 
computer gegeven, zodat deze een lopend programma kan onderbreken. Zou het mogelijk 
zijn deze beide IC's helemaal weg te laten en toch een goed werkende ontvanger over te 
houden? zo vroeg ik me af. De functies die deze IC's vervullen zouden dan door de 
software overgenomen moeten worden. Na enig puzzelen bleek dat inderdaad te kunnen, 
vooral omdat de microprocessor zelf ook al over een nauwkeurig kloksignaal beschikt. 
Deze klok loopt wel veel sneller dan die van de infrarood zender, maar er bestaat een 
vaste verhouding tussen deze kloksnelheden. In het programma kan daarom de lage 
kloksnelheid door aftellen uit de snelle klok afgeleid worden. Al met al wordt de 
infrarood ontvanger nu wel heel eenvoudig, zoals uit figuur 1 blijkt.  
 
Aansluiten 
 
De ontvanger kan direct vastgesoldeerd worden op een negenpolige D-connector die past 
op de joystickingang van een MSX-computer. Maar u mag er natuurlijk ook een stukje 
kabel tussen zetten. Zoals gebruikelijk sluiten we de ontvanger aan op joystickpoort 2, 
omdat op ingang 1 meestal al een muis aangesloten is. De weerstand en condensator 
vormen een door de fabrikant aanbevolen filter voor de 5 Volt voeding van de SFH505A. 
Het blijkt dat de zeer gevoelige ontvanger gestoord kan worden door de computer als we 
dit filter weglaten. De ontvanger gebruikt overigens minder dan één mA. Gloeilamplicht 
dat direct op de infrarood detector schijnt veroorzaakt geen problemen, maar het licht van 
TL-buizen of spaarlampen kan de infrarood ontvangst storen. 
 
De RC-5 code 
 
Voor degenen die niet zo goed bekend zijn met de RC-5 code volgt hier wat meer uitleg. 
Deze code is een door Philips ontwikkelde standaard voor infrarood afstandbediening van 
TV, radio, videorecorder, CD-speler en dergelijke toestellen. De RC-5 code wordt vooral 
door Europese fabrikanten van elektronica toegepast; veel oosterse merken hebben weer 
andere coderingen bedacht. Een afstandbediening met RC-5 code zendt voor elke 
ingedrukte toets een 14-bits codewoord uit, te beginnen met bit 13 en eindigend met bit 0. 
 
Opbouw RC-5 codewoord 
 
Bit 13-12: 2startbits  
Bit 11:      1 wisselbit  
Bit 10-6:    5 bits systeemcode  
Bit 5-0:     6 bits toetscode 
 
Het uitgezonden codewoord begint altijd met de twee startbits '1', waarna het wisselbit 
volgt. Dat is afwisselend 'O' en '1', elke keer wanneer een toets ingedrukt wordt. Op die 



manier kunnen we zien of twee opeenvolgende codewoorden, als ze dezelfde systeem- en 
toetscode hebben, het gevolg zijn van het wat langer vasthouden danwei van het twee 
keer indrukken van de toets. Zo voorkomt men dat u op uw televisie programma 22 krijgt 
in plaats van programma 2 als u toets 2 te lang indrukte. De systeemcode geeft aan voor 
welk toestel de verzonden code bestemd is. Er zijn met de vijf beschikbare bits 32 
verschillende mogelijkheden, waarvan een aantal in reserve wordt gehouden voor de 
toekomst. 
 
RC-5 systeemcodes 
 
0,1 Televisie 
4 Laserdisk 
5,6 Videorecorder 
8 Satellietontvanger 
9 Videocamera 
12 Compact Disk video 
17 Tuner 
18,23 Cassettespeler 
20 Compact Disk audio 
21 Platenspeler 
24-26 Toetsenbord 
29,30 Verlichting 
 
De zes bits toetscode die tenslotte volgt maakt het mogelijk 64 verschillende opdrachten 
te geven. De codes O t/m 9 worden daarbij altijd voor de overeenkomstige cijfers 
gebruikt. De overige toetscodes zijn merendeels afhankelijk van de gekozen 
systeemcode. Zo kent de televisie een aantal opdrachten voor het kiezen van teletekst-
pagina's, terwijl we voor een videorecorder de band met verschillende snelheden vooruit 
en terug kunnen laten lopen. 
 
Codering 
 
Het 14-bits RC-5 codewoord wordt in ca. 25 msec uitgezonden, dat is 1,8 msec per bit. 
Nu is het niet zo dat voor een '1' wel en voor '0' géén infrarood signaal wordt 
uitgezonden. Dat zou tot fouten aanleiding kunnen geven, omdat bij een korte 
onderbreking van de lichtstraal dan immers ook de bits '1' als een 'O' gedetecteerd zouden 
worden. In de RC-5 code is er op een eenvoudige maar slimme manier voor gezorgd, dat 
gecontroleerd kan worden of alle 14 bits van het codewoord inderdaad ontvangen zijn. 
Hierbij wordt een bit '1' gekenmerkt door het inschakelen van het infrarood signaal aan 
het begin van de bittijd. Evenzo detecteren we een 'O' aan het uitschakelen van het 
infrarood signaal op het juiste moment. Volgt op een bit '1' een 'O' dan is het simpel: het 
infrarood signaal wordt ingeschakeld voor de '1' en na 1,8 msec weer uitgeschakeld voor 
de 'O'. Iets ingewikkelder wordt het wanneer twee of meer dezelfde bits elkaar opvolgen, 
zoals de beide startbits '1' aan het begin van het codewoord. Het infrarood signaal wordt 
ingeschakeld voor het eerste startbit, maar moet halverwege de bittijd weer uitgeschakeld 
worden, zodat het na 1,8 msec opnieuw ingeschakeld kan worden voor het tweede 



startbit. Voor twee of meer opeenvolgende bits 'O' gebeurt precies het omgekeerde: 
hierbij wordt halverwege de bittijd het infrarood signaal juist ingeschakeld en dan aan het 
einde daarvan weer uitgeschakeld om zo de volgende 'O' aan te geven. 

Fig. 2: voorbeeld van een RC-5 code 
 
Decodering 
 
Deze codering, die tweefasencode wordt genoemd, is dus gekenmerkt door het in- of 
uitschakelen van het infraroodsignaal voor elk van de 14 bits uit het codewoord. 
Inschakelen betekent dat er een '1' volgt, uitschakelen geeft een 'O' aan. Het decoderen is 
dan ook niet moeilijk, maar daarbij moeten wel de overgangen halverwege de bittijd, die 
optreden als gelijke bits na elkaar verzonden worden, overgeslagen worden. Daarvoor 
hebben we een kloksignaal nodig, dat aangeeft wanneer het tijd is voor het 
volgende bit. In het schema in PRINT 39 werd dit kloksignaal gereconstrueerd met 
behulp van een tweetal IC's. In het nieuwe schema zijn die weggelaten en wordt het in de 
software opgelost. Als je er even goed over nadenkt is dat ook helemaal niet zo moeilijk. 
We weten al dat het RC-5 codewoord begint het eerste startbit, dat altijd '1' is en waarbij 
dus het infrarood signaal wordt ingeschakeld. Het volgende bit komt zo'n 1,8 msec later. 
Wordt het signaal al eerder uitgeschakeld, dan telt die verandering niet mee. Wel moeten 
we rekening houden met kleine verschillen in de kloksnelheid van verschillende 
afstandbedieningen. Daarom laten we het programma vanaf het begin van het eerste 
startbit ongeveer 1,5 msec wachten. De tussenovergang in het codewoord wordt daarmee 
in elk geval overgeslagen, want die komt al na zo'n 0,9 msec. Vervolgens wacht het 
programma op de eerstvolgende overgang. Als die tijdig komt is het een echte 
bitovergang, zodat daarvan bepaald kan worden of het een '1 of een 'O' betreft. Daarna 
laten we opnieuw de wachttijd ingaan. Dit gaat zo door tot er precies 14 bits gevonden 
zijn. De wachtlussen in het machinetaalprogramma gelden voor een MSX-computer die 
op de standaard kloksnelheid loopt. Wie zijn computer omgebouwd heeft naar een hogere 
kloksnelheid zal deze moeten terugschakelen of de wachtlussen moeten aanpassen. 
 
Programma 
 
Het hierbij afgedrukte MSX BASIC-programma is uit drie delen opgebouwd. Het eerste 
deel zorgt voor het installeren van het stukje machinetaal, waarmee de ontvangen 
infrarood signalen gedetecteerd en gedecodeerd worden. De decodering is afgestemd op 
de RC-5 code en werkt dan ook niet voor vreemde coderingen! In het tweede deel van het 
programma wordt deze machinetaal-routine via een USR-opdracht aangeroepen in 
afwachting van het te ontvangen codewoord. Om geen informatie verloren te laten gaan 



wordt de interrupt uitgeschakeld, waardoor deze opdracht niet via het toetsenbord van de 
computer af te breken is. Als er helemaal geen infrarood signaal ontvangen wordt, zou de 
computer daar eeuwig op blijven wachten. Daarom bevat het machinetaal-programma 
een maximale wachttijd, die u met het getal Z zelf kunt instellen. Deze wachttijd bedraagt 
ongeveer Z/3 seconden; met de Z=33 in het programma is dat dus zo'n tien seconden. 
Wordt gedurende de wachttijd géén infrarood signaal ontvangen, dan keert de routine 
terug naar BASIC en drukt de boodschap "Geen code ontvangen" af. Als er wel een 
codewoord ontvangen is dan worden het wisselbit W, de systeemcode S en de toetscode 
T uit het gedecodeerde signaal bepaald. 
 
Grafische weergave 
 
Het derde deel van het BASIC-programma toont een grafische weergave van de 
ontvangen infrarood signalen. Dit is mogelijk omdat het nog niet gedecodeerde 
bitpatroon tijdens de ontvangst in het computergeheugen opgeslagen wordt. Het is te 
vinden direct na het stukje machinetaal, vanaf adres &HD054. Zo krijgt u een grafische 
weergave te zien, ook als het ontvangen signaal geen RC-5 code betreft. U kunt daarmee 
in elk geval de werking van uw afstandbediening controleren. Als het bitpatroon van een 
vreemde code overigens veel langer dan 25 msec duurt, zal het niet helemaal op het 
scherm passen. Dit programmagedeelte bevat in regel 270 nog een extra test op juistheid 
van de ontvangen code. Volgens RC-5 moeten bit 13 en bit 12, dat zijn de twee startbits, 
'1' zijn. De ontvangen code moet dus minimaal de waarde 2 A 13 -t- 2 A 12 = 12288 
hebben. Als dat niet zo is krijgt u de boodschap 
"Dit is geen RC-5 code!". Het gedecodeerde wisselbit, de systeemcode en de toetscode 
krijgt u alleen te zien als een echt RC-5 codewoord ontvangen werd. U kunt de hier 
beschreven infrarood ontvanger gebruiken om functies in uw eigen programma aan te 
sturen. In dat geval kan het derde deel van het hier afgedrukte programma uiteraard 
weggelaten worden. De in het tweede deel berekende codes kunnen direct gebruikt 
worden om uit verschillende mogelijkheden te kiezen. Wel is het raadzaam de RC-5 test 
uit regel 270 toe te passen, bijvoorbeeld als programmaregel 175. 
 
In PRINT 38 van juli 1990 staat de beschrijving van een infrarood zender volgens de RC-
5 code. Daarmee kunt u via uw MSX-computer uw televisie, videorecorder en andere 
toestellen aansturen, voor zover deze tenminste voorzien zijn van een infrarood ontvanger 
met RC-5 code. Het schema is erg eenvoudig en bevat slechts één IC type NE555, dat 
nog geen gulden kost. Er hoeft niets afgeregeld te worden. Het nabouwen levert dan ook 
voor zover bekend geen problemen op. Om het u nog eenvoudiger te maken heeft Theo 
Maassen een printlayout met componentenopstelling gemaakt, die u hierbij afgedrukt 
ziet. Voor verdere gegevens zie PRINT 38, blz 19 en 20. 



 



 
10     'Infrarood-detector voor RC-5 code met SFH 505A 
20     'Detectoruitgang op pen 7 van joystickpoort 2 
30     '(c) Peter van Overbeek, October 1993 
40     DEFINT A-Z: A=&HD000: C=9574: DEFUSR=A 
50     FOR I=0 TO 83: READ D$: D=VAL("&H"+D$):POKE A+I,D 
60       C=C-D: NEXT: IF C THEN PRINT"FOUT IN DATA": END 
70     DATA F3,21,54,D0,11,FF,FF,D5,DD,E1,1B,3A,F8,F7,3C,47 
80     DATA10,FE,7A,B3,28,39,DB,A2,E6,20,20,EE,DD,23,11,E0 
90     DATA 01,DB,A2,E6,20,77,06,0A,10,FE,23,1B,7A,B3,20,F1 
100    DATA FB,21,53,D0,06,0E,0E,10,DD,29,7E,23,BE,20,05,0D 
110    DATA 20,F8,18,0B,B7,28,02,DD,23,11,16,00,19,10,E7,DD 
120    DATA 22,F8,F7,C9 
130    : 
140    'Ontvangen en decoderen van een RC-5 codewoord 
150    CLS: PRINT"Druk 'n toets op de afstandbediening" 
160    Z=33 ' Z is een maat voor de zaptijd 
170    C=USR(Z): IF C=-L THEN PRINT"Geen codeontvangen": END 
180    W=(C\2^11) MOD 2 ' W is het wisselbit (0 of 1) 
190    S=(C\64) MOD 32   ' S is de systeemcode (0..31) 
200    T=C MOD 64         ' T is de toetscode (0..63) 
210    : 
220    'Grafische weergave van de ontvangen RC-5 code 
230    SCREEN7: LINE (0,32)-(30,32): OPEN"GRP:" AS#1 
240    A=&HD054: FOR X=0 TO 479: Y=A+X: PSET(X+31,PEEK(Y)) 
250    IF PEEK(Y) <> PEEK(Y-L) THEN LINE (X+31,0)-(X+31,32) 
260    NEXT :B=12288: PSET (0,50) 
270    IF C<12288 THEN PRINT#1,"Dit is geen RC-5 code!": GOTO 300 
280    PRINT#1,"Ontvangen code: &B";BIN$(C);"=";C 
290    PRINT#1, "Wisselbit:" ;W; "Systeemcode:" ;S;"Toetscode:";T 
300    PSET(9,200): PRINT#1,"Druk 'n toets ,ESC=stoppen" 
310    I$=INPUT$(1): IF I$=CHR$(27) THEN END 
320    SCREEN 0: CLOSE#1: GOTO 140 

